Pensieve header: Invariance proof by Roland.

Proof of invariance under Reidemeister 3

The program to compute the ρ invariant.

```
In[∘]:= ρ[K_] := ρ[K] = Module[{Cs, r, n, B, A, c, s, i, j, Δ, G, g, ρ1},
    {Cs, r} = List @@ RVK[K]; n = Length[Cs]; B = Table[0, 2 n, 2 n + 1];
    Do[{s, i, j} = c;
     B[[{i, j}, {i, j, i + 1, j + 1}]] = ( 1 0  -1      0   )
                                          ( 0 1  T^s - 1 -T^s ), {c, Cs}];
    A = B[[All, 2 ;;]];
    Δ = T^((Total[r]-Total[First/@Cs])/2) Det[A];
    G = Prepend[Table[0, 2 n]][Inverse[A]]; g_{α_,β_} := G[[α, β]];
    ρ1 = Δ^2 Sum[{s, i, j} = c;
       s ((1 - T^s) g_ij (g_ij - g_jj) + 2 g_ii g_ij - g_ij g_ji - g_ii g_jj - g_ij + g_jj - 1 / 2), {c, Cs}];
    ρ1 += Δ^2 Sum[r[[k]] (g_kk - 1 / 2), {k, 2 n}] ;
    Factor@{Δ, ρ1}];
```

```
δ_{i_,j_} := If[Simplify[i] === Simplify[j], 1, 0] (*Kronecker delta*)
```

We start by deriving four relations on the matrix entries of G. These should be derivable directly from $A^{-1} A = I$ using the fact that we know B explicitly.
I however derived them from the Heisenberg commutation relations. In these equations {i,j} stands for the overpass and underpass at the same crossing, α can be arbitrary.

```
rule1i_{α_,i_,j_} := g_{i+1,α} → g_{i,α} - δ_{i,α}
rule1j_{α_,i_,j_} := g_{j+1,α} → g_{j,α} - δ_{j,α} + (1 - T^-1) (g_{i,α} - δ_{i,α} - g_{j,α} + δ_{j,α})
rule2i_{α_,i_,j_} := g_{α,i+1} → g_{α,i} + (1 - T) g_{α,j} + δ_{α,i+1}
rule2j_{α_,i_,j_} := g_{α,j+1} → T g_{α,j} + δ_{α,j+1}
```

We encode the edges near the Reidemeister 3 move as shown:



The crossings in the left hand side are listed in LXs and those of the righthand picture are in RXs. The format of a crossing is to list the overpass first. All crossings are positive.

```
In[]:= LXs = {{a, c + 1}, {b, c}, {a + 1, b + 1}};
RXs = {{a, b}, {a + 1, c}, {b + 1, c + 1}};
```

Next use the above relations to generate rewriting rules expressing $g_{x,y}$ in terms of $g_{a|b|c,a|b|c}$

```
In[]:= LRules = Flatten@Table[{i, j} = cr;
    {rule1i[k,i,j], rule1j[k,i,j], rule2i[k,i,j], rule2j[k,i,j]}
    , {cr, LXs}, {k, {a, b, c, a + 1, b + 1, c + 1}}];
RRules = Flatten@Table[{i, j} = cr;
    {rule1i[k,i,j], rule1j[k,i,j], rule2i[k,i,j], rule2j[k,i,j]}
    , {cr, RXs}, {k, {a, b, c, a + 1, b + 1, c + 1}}];
```

According to the above program the weight assigned to each crossing {i,j} is given as below.

In[]:= $\text{CrossingWeight}_{i\_,j\_} := (1 - T)\, g_{i,j} \left(g_{i,j} - g_{j,j}\right) + 2\, g_{i,i}\, g_{i,j} - g_{i,j}\, g_{j,i} - g_{i,i}\, g_{j,j} - g_{i,j} + g_{j,j} - 1/2$

Assuming that the matrix entries $g_{x,y}$ from the LHS and the RHS are equal when neither of x,y are in {a+1,b+1,c+1}
we can thus proceed to check invariance under Reidemeister 3 directly as below.

In[∘]:=
```
R3LHS = Sum[{i, j} = cr; CrossingWeight_{i,j}, {cr, LXs}]
R3RHS = Sum[{i, j} = cr;
  CrossingWeight_{i,j}, {cr, RXs}]
```

Out[∘]=
$$-\frac{3}{2} - g_{a,1+c} + 2\, g_{a,a}\, g_{a,1+c} - g_{1+a,1+b} + 2\, g_{1+a,1+a}\, g_{1+a,1+b} - g_{b,c} + 2\, g_{b,b}\, g_{b,c} - g_{1+a,1+b}\, g_{1+b,1+a} + (1-T)\, g_{1+a,1+b}\, (g_{1+a,1+b} - g_{1+b,1+b}) + g_{1+b,1+b} - g_{1+a,1+a}\, g_{1+b,1+b} - g_{b,c}\, g_{c,b} + (1-T)\, g_{b,c}\, (g_{b,c} - g_{c,c}) + g_{c,c} - g_{b,b}\, g_{c,c} - g_{a,1+c}\, g_{1+c,a} + (1-T)\, g_{a,1+c}\, (g_{a,1+c} - g_{1+c,1+c}) + g_{1+c,1+c} - g_{a,a}\, g_{1+c,1+c}$$

Out[∘]=
$$-\frac{3}{2} - g_{a,b} + 2\, g_{a,a}\, g_{a,b} - g_{1+a,c} + 2\, g_{1+a,1+a}\, g_{1+a,c} - g_{a,b}\, g_{b,a} + (1-T)\, g_{a,b}\, (g_{a,b} - g_{b,b}) + g_{b,b} - g_{a,a}\, g_{b,b} - g_{1+b,1+c} + 2\, g_{1+b,1+b}\, g_{1+b,1+c} - g_{1+a,c}\, g_{c,1+a} + (1-T)\, g_{1+a,c}\, (g_{1+a,c} - g_{c,c}) + g_{c,c} - g_{1+a,1+a}\, g_{c,c} - g_{1+b,1+c}\, g_{1+c,1+b} + (1-T)\, g_{1+b,1+c}\, (g_{1+b,1+c} - g_{1+c,1+c}) + g_{1+c,1+c} - g_{1+b,1+b}\, g_{1+c,1+c}$$

In[∘]:=
```
(R3LHS //. LRules) - (R3RHS //. RRules) // Simplify
```

Out[∘]= 0